SOLUCIÓN

FACULTAD DE INGENIERÍA
ECUACIONES DIFERENCIALES
SEGUNDO EXAMEN FINAL
SEMESTRE 2015-2

2015 JUNIO 5

> *restart*

1) Resolver la ecuación diferencial

> $\dfrac{(1-\sin(x))}{y(x)} \cdot diff(y(x), x) - \log(y(x)) = 0$

$$\frac{(1-\sin(x))\left(\frac{d}{dx}\, y(x)\right)}{y(x)} - \ln(y(x)) = 0 \qquad \textbf{(1)}$$

RESPUESTA 1)

> $Ecuacion := \dfrac{(1-\sin(x))}{y(x)} \cdot diff(y(x), x) - \log(y(x)) = 0$

$$Ecuacion := \frac{(1-\sin(x))\left(\frac{d}{dx}\, y(x)\right)}{y(x)} - \ln(y(x)) = 0 \qquad \textbf{(2)}$$

> $Solucion := simplify(dsolve(Ecuacion))$

$$Solucion := y(x) = e^{e^{-\frac{2 + \_C1 \tan\left(\frac{1}{2}x\right) - \_C1}{\tan\left(\frac{1}{2}x\right) - 1}}} \qquad \textbf{(3)}$$

> $with(DEtools)$:

> $odeadvisor(Ecuacion)$

$$[\_separable] \qquad \textbf{(4)}$$

> $M := -\log(y);\ N := \dfrac{(1-\sin(x))}{y}$

$$M := -\ln(y)$$
$$N := \frac{1-\sin(x)}{y} \qquad \textbf{(5)}$$

> $P := -1;\ Q := \log(y);\ R := 1 - \sin(x);\ S := \dfrac{1}{y}$

$$P := -1$$
$$Q := \ln(y)$$
$$R := 1 - \sin(x)$$
$$S := \frac{1}{y} \qquad \textbf{(6)}$$

> $SolUno := Int\left(\dfrac{P}{R}, x\right);\ SolDos := Int\left(\dfrac{S}{Q}, y\right)$

$$SolUno := \int \left(-\frac{1}{1-\sin(x)} \right) dx$$

$$SolDos := \int \frac{1}{\ln(y)\, y}\, dy \tag{7}$$

> $SolucionGeneral := int\left(\frac{P}{R}, x\right) + int\left(\frac{S}{Q}, y\right) = C_1$

$$SolucionGeneral := \frac{2}{\tan\left(\frac{1}{2}x\right) - 1} + \ln(\ln(y)) = C_1 \tag{8}$$

> $Sol := \frac{2}{\tan\left(\frac{1}{2}x\right) - 1} + \ln(\ln(y(x))) = C_1$

$$Sol := \frac{2}{\tan\left(\frac{1}{2}x\right) - 1} + \ln(\ln(y(x))) = C_1 \tag{9}$$

comprobación

> $DerSol := simplify(isolate(diff(Sol, x), diff(y(x), x)))$

$$DerSol := \frac{d}{dx} y(x) = -\frac{\ln(y(x))\, y(x)}{-1 + \sin(x)} \tag{10}$$

> $DerOrig := isolate(Ecuacion, diff(y(x), x))$

$$DerOrig := \frac{d}{dx} y(x) = \frac{\ln(y(x))\, y(x)}{1 - \sin(x)} \tag{11}$$

> $Comprobacion := simplify(rhs(DerSol) - rhs(DerOrig)) = 0$

$$Comprobacion := 0 = 0 \tag{12}$$

FIN RESPUESTA 1)

> *restart*

2) Obtener la solución general de la ecuación diferencial

> $y''''' + 2\cdot y'''' + 4\cdot y''' - 2\cdot y'' - 5\cdot y' = 10\cdot x - 1$

$$\frac{d^5}{dx^5} y(x) + 2\left(\frac{d^4}{dx^4} y(x)\right) + 4\left(\frac{d^3}{dx^3} y(x)\right) - 2\left(\frac{d^2}{dx^2} y(x)\right) - 5\left(\frac{d}{dx} y(x)\right) = 10x - 1 \tag{13}$$

RESPUESTA 2)

> $Ecuacion := \frac{d^5}{dx^5} y(x) + 2\left(\frac{d^4}{dx^4} y(x)\right) + 4\left(\frac{d^3}{dx^3} y(x)\right) - 2\left(\frac{d^2}{dx^2} y(x)\right) - 5\left(\frac{d}{dx} y(x)\right) = 10x - 1$

$$Ecuacion := \frac{d^5}{dx^5} y(x) + 2\left(\frac{d^4}{dx^4} y(x)\right) + 4\left(\frac{d^3}{dx^3} y(x)\right) - 2\left(\frac{d^2}{dx^2} y(x)\right) - 5\left(\frac{d}{dx} y(x)\right) = 10x - 1 \tag{14}$$

> $Solucion := simplify(dsolve(Ecuacion))$

$$Solucion := y(x) = -x^2 + \_C1\, e^{x} - \_C2\, e^{-x} - \frac{1}{5}\,\_C3\, e^{-x}\cos(2x) + \frac{2}{5}\,\_C3\, e^{-x}\sin(2x) \tag{15}$$

$$-\frac{2}{5}\,\_C4\,e^{-x}\cos(2\,x)-\frac{1}{5}\,\_C4\,e^{-x}\sin(2\,x)+x+\_C5$$

> $EcuacionHom := lhs(Ecuacion) = 0$

$$EcuacionHom := \frac{d^5}{dx^5}\,y(x)+2\left(\frac{d^4}{dx^4}\,y(x)\right)+4\left(\frac{d^3}{dx^3}\,y(x)\right)-2\left(\frac{d^2}{dx^2}\,y(x)\right)-5\left(\frac{d}{dx}\,y(x)\right)=0 \tag{16}$$

> $Q := rhs(Ecuacion)$

$$Q := 10\,x-1 \tag{17}$$

> $EcuacionCarac := m^5+2\cdot m^4+4\cdot m^3-2\cdot m^2-5\cdot m=0$

$$EcuacionCarac := m^5+2\,m^4+4\,m^3-2\,m^2-5\,m=0 \tag{18}$$

> $Raiz := solve(EcuacionCarac)$

$$Raiz := 0,\ 1,\ -1,\ -1+2\,\mathrm{I},\ -1-2\,\mathrm{I} \tag{19}$$

> $SolUno := y(x)=\exp(Raiz_1\cdot x);\ SolDos := y(x)=\exp(Raiz_2\cdot x);\ SolTres := y(x)=\exp(Raiz_3\cdot x);\ SolCuatro := y(x)=\exp(\mathrm{Re}(Raiz_4)\cdot x)\cdot\cos(\mathrm{Im}(Raiz_4)\cdot x);\ SolCinco := y(x)=\exp(\mathrm{Re}(Raiz_4)\cdot x)\cdot\sin(\mathrm{Im}(Raiz_4)\cdot x)$

$$SolUno := y(x)=1$$

$$SolDos := y(x)=e^{x}$$

$$SolTres := y(x)=e^{-x}$$

$$SolCuatro := y(x)=e^{-x}\cos(2\,x)$$

$$SolCinco := y(x)=e^{-x}\sin(2\,x) \tag{20}$$

> $with(linalg):$

> $WW := wronskian([rhs(SolUno), rhs(SolDos), rhs(SolTres), rhs(SolCuatro), rhs(SolCinco)], x)$

$$WW := \begin{bmatrix} 1 & e^{x} & e^{-x} & e^{-x}\cos(2\,x) & e^{-x}\sin(2\,x) \\ 0 & e^{x} & -e^{-x} & -e^{-x}\cos(2\,x)-2\,e^{-x}\sin(2\,x) & -e^{-x}\sin(2\,x)+2\,e^{-x}\cos(2\,x) \\ 0 & e^{x} & e^{-x} & -3\,e^{-x}\cos(2\,x)+4\,e^{-x}\sin(2\,x) & -3\,e^{-x}\sin(2\,x)-4\,e^{-x}\cos(2\,x) \\ 0 & e^{x} & -e^{-x} & 11\,e^{-x}\cos(2\,x)+2\,e^{-x}\sin(2\,x) & 11\,e^{-x}\sin(2\,x)-2\,e^{-x}\cos(2\,x) \\ 0 & e^{x} & e^{-x} & -7\,e^{-x}\cos(2\,x)-24\,e^{-x}\sin(2\,x) & -7\,e^{-x}\sin(2\,x)+24\,e^{-x}\cos(2\,x) \end{bmatrix} \tag{21}$$

> $BB := array([0, 0, 0, 0, Q])$

$$BB := \begin{bmatrix} 0 & 0 & 0 & 0 & 10\,x-1 \end{bmatrix} \tag{22}$$

> $SOL := linsolve(WW, BB) : Aprima := SOL_1;\ Bprima := SOL_2;\ Dprima := SOL_3;\ Eprima := simplify(SOL_4);\ Fprima := simplify(SOL_5)$

$$Aprima := -2\,x+\frac{1}{5}$$

$$Bprima := \frac{1}{16}\,\frac{10\,x-1}{e^{x}}$$

$$Dprima := \frac{1}{8}\,\frac{10\,x-1}{e^{-x}}$$

$$Eprima := \frac{1}{80}\,(10\,x-1)\,e^{x}\,(-3\sin(2\,x)+\cos(2\,x))$$

$$Fprima := \frac{1}{80}\,(10\,x-1)\,e^{x}\,(3\cos(2\,x)+\sin(2\,x)) \qquad \textbf{(23)}$$

> $A := simplify(int(Aprima, x)) + C_1;\ B := simplify(int(Bprima, x)) + C_2;\ DD := simplify(int(Dprima, x)) + C_3;\ E := simplify(int(Eprima, x)) + C_4;\ F := simplify(int(Fprima, x)) + C_5$

$$A := -x^2 + \frac{1}{5}\,x + C_1$$

$$B := -\frac{1}{16}\,(9+10\,x)\,e^{-x} + C_2$$

$$DD := \frac{1}{8}\,(10\,x-11)\,e^{x} + C_3$$

$$E := \frac{1}{80}\,e^{x}\left(-4\cos(x)\sin(x)\,x - 10\cos(x)\sin(x) - 14\,x + 5 + 28\cos(x)^2\,x - 10\cos(x)^2\right) + C_4$$

$$F := \frac{1}{80}\,e^{x}\left(28\cos(x)\sin(x)\,x - 10\cos(x)\sin(x) - 2\,x - 5 + 4\cos(x)^2\,x + 10\cos(x)^2\right) + C_5 \qquad \textbf{(24)}$$

> $SolucionGeneral := y(x) = simplify(expand(A\cdot rhs(SolUno) + B\cdot rhs(SolDos) + DD\cdot rhs(SolTres) + E\cdot rhs(SolCuatro) + F\cdot rhs(SolCinco)))$

$$SolucionGeneral := y(x) = -\left(2\,e^{x} + x^2\,e^{x} - e^{x}\,x - 2\,C_5\cos(x)\sin(x) - C_1\,e^{x} - 2\,C_4\cos(x)^2 - e^{2\,x}\,C_2 - C_3 + C_4\right)e^{-x} \qquad \textbf{(25)}$$

> $Solucion$

$$y(x) = -x^2 + \_C1\,e^{x} - \_C2\,e^{-x} - \frac{1}{5}\,\_C3\,e^{-x}\cos(2\,x) + \frac{2}{5}\,\_C3\,e^{-x}\sin(2\,x) - \frac{2}{5}\,\_C4\,e^{-x}\cos(2\,x) - \frac{1}{5}\,\_C4\,e^{-x}\sin(2\,x) + x + \_C5 \qquad \textbf{(26)}$$

> $comprobacion_1 := simplify(eval(subs(y(x) = rhs(SolucionGeneral), lhs(Ecuacion) - rhs(Ecuacion) = 0)))$

$$comprobacion_1 := 0 = 0 \qquad \textbf{(27)}$$

> $comprobacion_2 := simplify(eval(subs(y(x) = rhs(Solucion), lhs(Ecuacion) - rhs(Ecuacion) = 0)))$

$$comprobacion_2 := 0 = 0 \qquad \textbf{(28)}$$

FIN RESPUESTA 2)

> *restart*

3) Determinar el sistema equivalente de primer orden, en forma matricial, de la ecuación diferencial

> $y''' - 4\,y = 2\,\cdot\sin(t)$

$$\frac{d^3}{dx^3}\,y(x) - 4\,y(x) = 2\,\sin(t) \tag{29}$$

RESPUESTA 3)

> $Ecuacion := \frac{d^3}{dx^3}\,y(x) - 4\,y(x) = 2\,\sin(t)$

$$Ecuacion := \frac{d^3}{dx^3}\,y(x) - 4\,y(x) = 2\,\sin(t) \tag{30}$$

> $Q := rhs(Ecuacion)$

$$Q := 2\,\sin(t) \tag{31}$$

> $Sistema := diff\,(y_1(t), t) = y_2(t), diff\,(y_2(t), t) = y_3(t), diff\,(y_3(t), t) = 4\cdot y_1(t) + 2\cdot\sin(t)$:
> $Sistema_1$; $Sistema_2$; $Sistema_3$

$$\frac{d}{dt}\,y_1(t) = y_2(t)$$

$$\frac{d}{dt}\,y_2(t) = y_3(t)$$

$$\frac{d}{dt}\,y_3(t) = 4\,y_1(t) + 2\,\sin(t) \tag{32}$$

> $AA := array([[0, 1, 0], [0, 0, 1], [4, 0, 0]])$

$$AA := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 4 & 0 & 0 \end{bmatrix} \tag{33}$$

> $BB := array([0, 0, Q])$

$$BB := \begin{bmatrix} 0 & 0 & 2\,\sin(t) \end{bmatrix} \tag{34}$$

> $DerX := array\left(\left[diff\,(x_1(t), t), diff\,(x_2(t), t), diff\,(x_3(t), t)\right]\right)$

$$DerX := \begin{bmatrix} \frac{d}{dt}\,x_1(t) & \frac{d}{dt}\,x_2(t) & \frac{d}{dt}\,x_3(t) \end{bmatrix} \tag{35}$$

> $X := array\left(\left[x_1(t), x_2(t), x_3(t)\right]\right)$

$$X := \begin{bmatrix} x_1(t) & x_2(t) & x_3(t) \end{bmatrix} \tag{36}$$

> $Sist := evalm(evalm(AA \,\&*\, X) + BB)$: $DerX_1 = Sist_1$; $DerX_2 = Sist_2$; $DerX_3 = Sist_3$

$$\frac{d}{dt}\,x_1(t) = x_2(t)$$

$$\frac{d}{dt}\,x_2(t) = x_3(t)$$

$$\frac{d}{dt}\,x_3(t) = 4\,x_1(t) + 2\,\sin(t) \tag{37}$$

FIN RESPUESTA 3)

> *restart*

4) Resolver la ecuacion diferencial

> $y''(t) - 4 \cdot y'(t) + 3 \cdot y(t) = \mathrm{Dirac}(t-1); y(0) = 0; \mathrm{D}(y)(0) = 0$

$$\mathrm{D}^{(2)}(y)(t) - 4\,\mathrm{D}(y)(t) + 3\,y(t) = \mathrm{Dirac}(t-1)$$
$$y(0) = 0$$
$$\mathrm{D}(y)(0) = 0 \qquad \textbf{(38)}$$

RESPUESTA 4)

> $Ecuacion := y''(t) - 4 \cdot y'(t) + 3 \cdot y(t) = \mathrm{Dirac}(t-1); Condiciones := y(0) = 0, \mathrm{D}(y)(0) = 0$

$$Ecuacion := \mathrm{D}^{(2)}(y)(t) - 4\,\mathrm{D}(y)(t) + 3\,y(t) = \mathrm{Dirac}(t-1)$$
$$Condiciones := y(0) = 0, \mathrm{D}(y)(0) = 0 \qquad \textbf{(39)}$$

> $Solucion := dsolve(\{Ecuacion, Condiciones\})$

$$Solucion := y(t) = \frac{1}{2}\,\mathrm{Heaviside}(t-1)\,\left(\mathrm{e}^{-3+3t} - \mathrm{e}^{t-1}\right) \qquad \textbf{(40)}$$

> $with(inttrans):$

> $TLEcuacion := subs(Condiciones, laplace(Ecuacion, t, s))$

$$TLEcuacion := s^2\,laplace(y(t), t, s) - 4\,s\,laplace(y(t), t, s) + 3\,laplace(y(t), t, s) = \mathrm{e}^{-s} \qquad \textbf{(41)}$$

> $TLSolucion := isolate(TLEcuacion, laplace(y(t), t, s))$

$$TLSolucion := laplace(y(t), t, s) = \frac{\mathrm{e}^{-s}}{s^2 - 4\,s + 3} \qquad \textbf{(42)}$$

> $SolucionParticular := invlaplace(TLSolucion, s, t)$

$$SolucionParticular := y(t) = \mathrm{Heaviside}(t-1)\,\mathrm{e}^{2t-2}\,\sinh(t-1) \qquad \textbf{(43)}$$

FIN RESPUESTA 4)

> *restart*

5) Obtener la inversa de Laplace

> $F := \frac{s^2}{(s^2+9)^2}$

$$F := \frac{s^2}{(s^2+9)^2} \qquad \textbf{(44)}$$

RESPUESTA 5)

> $with(inttrans):$

> $f := invlaplace(F, s, t)$

$$f := \frac{1}{6}\,\sin(3\,t) + \frac{1}{2}\,t\cos(3\,t) \qquad \textbf{(45)}$$

> $G := \frac{s}{s^2+9}; H := \frac{s}{s^2+9}; G \cdot H$

$$G := \frac{s}{s^2+9}$$

$$H := \frac{s}{s^2+9}$$

$$\frac{s^2}{\left(s^2+9\right)^2} \quad \textbf{(46)}$$

> $g := invlaplace(G, s, t);\ h := invlaplace(H, s, t)$

$$g := \cos(3\,t)$$

$$h := \cos(3\,t) \quad \textbf{(47)}$$

por convolución

> $ff := Int(subs(t=\tau, g)\cdot subs(t=t-\tau, h), \tau=0..t) = int(subs(t=\tau, g)\cdot subs(t=t-\tau, h), \tau=0..t)$

$$ff := \int_0^t \cos(3\,\tau)\,\cos(3\,t-3\,\tau)\,\mathrm{d}\tau = \frac{1}{6}\sin(3\,t) + \frac{1}{2}\,t\cos(3\,t) \quad \textbf{(48)}$$

FIN RESPUESTA 5)

> *restart*

6) Obtener una solución completa de la ecuación diferencial en derivadas parciales, para una constante de separación igual a -1

> $t^2\cdot diff(u(x,t),x) - x^2\cdot diff(u(x,t),t) = 0$

$$t^2\left(\frac{\partial}{\partial x}u(x,t)\right) - x^2\left(\frac{\partial}{\partial t}u(x,t)\right) = 0 \quad \textbf{(49)}$$

RESPUESTA 6)

> $Ecuacion := t^2\cdot diff(u(x,t),x) = x^2\cdot diff(u(x,t),t)$

$$Ecuacion := t^2\left(\frac{\partial}{\partial x}u(x,t)\right) = x^2\left(\frac{\partial}{\partial t}u(x,t)\right) \quad \textbf{(50)}$$

> $SolucionGeneral := pdsolve(Ecuacion)$

$$SolucionGeneral := u(x,t) = \_F1\left(x^3+t^3\right) \quad \textbf{(51)}$$

> $EcuacionSeparable := eval(subs(u(x,t) = F(x)\cdot G(t), Ecuacion))$

$$EcuacionSeparable := t^2\left(\frac{\mathrm{d}}{\mathrm{d}x}F(x)\right)G(t) = x^2\,F(x)\left(\frac{\mathrm{d}}{\mathrm{d}t}G(t)\right) \quad \textbf{(52)}$$

> $EcuacionSeparada := \dfrac{lhs(EcuacionSeparable)}{t^2\cdot x^2\cdot F(x)\cdot G(t)} = \dfrac{rhs(EcuacionSeparable)}{t^2\cdot x^2\cdot F(x)\cdot G(t)}$

$$EcuacionSeparada := \frac{\frac{\mathrm{d}}{\mathrm{d}x}F(x)}{x^2\,F(x)} = \frac{\frac{\mathrm{d}}{\mathrm{d}t}G(t)}{t^2\,G(t)} \quad \textbf{(53)}$$

> $EcuacionX := lhs(EcuacionSeparada) = \alpha;\ EcuacionT := rhs(EcuacionSeparada) = \alpha$

$$EcuacionX := \frac{\frac{\mathrm{d}}{\mathrm{d}x}F(x)}{x^2\,F(x)} = \alpha$$

$$EcuacionT := \frac{\frac{d}{dt} G(t)}{t^2 G(t)} = \alpha \quad \textbf{(54)}$$

> *SolucionX* := *dsolve*(*subs*(alpha = −1, *EcuacionX*)); *SolucionT* := *dsolve*(*subs*(alpha = −1, *EcuacionT*));

$$SolucionX := F(x) = \_C1\, e^{-\frac{1}{3} x^3}$$

$$SolucionT := G(t) = \_C1\, e^{-\frac{1}{3} t^3} \quad \textbf{(55)}$$

> *SolucionCompleta* := *u*(*x*, *t*) = *rhs*(*SolucionX*) · *subs*(_C1 = 1, *rhs*(*SolucionT*))

$$SolucionCompleta := u(x, t) = \_C1\, e^{-\frac{1}{3} x^3}\, e^{-\frac{1}{3} t^3} \quad \textbf{(56)}$$

FIN RESPUESTA 6)

> *restart*

FIN DEL EXAMEN